NOAA TM ERL AL-6

A UNITE
DEPART
COMM
PUBLI

NOAA Technical Memorandum ERL AL-6

U.S. DEPARTMENT OF COMMERCE
NATIONAL OCEANIC AND ATMOSPHERIC ADMINISTRATION
Environmental Research Laboratories

Accurate Langmuir Probe Measurements With an On-Line Computer

K. H. GEISSLER
J. H. DARR

Aeronomy Laboratory
BOULDER, COLORADO
June 1973

NMENTAL RESEARCH LABORATORIES

AERONOMY LABORATORY

IMPORTANT NOTICE

Technical Memoranda are used to insure prompt dissemination of special studies which, though of interest to the scientific community, may not be ready for formal publication. Since these papers may later be published in a modified form to include more recent information or research results, abstracting, citing, or reproducing this paper in the open literature is not encouraged. Contact the author for additional information on the subject matter discussed in this Memorandum.

NATIONAL OCEANIC AND ATMOSPHERIC ADMINISTRATION

U.S. DEPARTMENT OF COMMERCE
National Oceanic and Atmospheric Administration
Environmental Research Laboratories

NOAA Technical Memorandum ERL AL-6

ACCURATE LANGMUIR PROBE MEASUREMENTS WITH AN ON-LINE COMPUTER

K. H. Geissler
J. H. Darr

Aeronomy Laboratory
Boulder, Colorado
June 1973

TABLE OF CONTENTS

ACCURATE LANGMUIR PROBE MEASUREMENTS WITH AN ON-LINE COMPUTER

K. H. Geissler and J. H. Darr

1. INTRODUCTION

Langmuir probe theory has made major advances in recent years, mainly through the rigorous application of trajectory theory (compare Laframboise and Parker, 1973; Parker and Whipple, Jr., 1967). However, certain difficulties remain on the experimental side, namely the significant discrepancy between ionospheric temperatures measured by rocket borne Langmuir probes and the temperatures obtained simultaneously from incoherent backscatter radar, as discussed by Brace and McClure (1970) and Carlson and Sayers (1970). A similar discrepancy was observed in laboratory plasmas, where Langmuir probes can be used both to measure the electron temperature from the cross spectra of thermal density fluctuations (Geissler et al., 1972), and also to measure the temperature simultaneously in the conventional way. Again, the temperature deduced from the density fluctuations tends to be lower. The connection to the ionospheric temperature measurement by backscatter radar is given by the fact that the density fluctuations which are detected by the radar are deduced from the same theory used in the laboratory fluctuation analysis. Hoegy (1971) has shown that the discrepancy in temperature measurements can be explained by assuming non-Maxwellian velocity distributions for the electrons.

This report is a study of a method for measuring Langmuir probe characteristics accurately enough to either verify or exclude the hypothesis of non-Maxwellian velocity distributions. Section 2 gives general considerations for the use of Langmuir probes. Section 3 reviews the equations used for evaluating the measured probe characteristics. Sections 4 and 5 outline the data accepting and reduction procedures used. Program listings are given in the Appendix.

2. GENERAL CONSIDERATIONS FOR THE USE OF LANGMUIR PROBES

A successful use of Langmuir probes requires the consideration of the following problems:

a) Perturbations introduced by the probe structure.

b) Surface effects on the probe or counter electrode.

c) Signal to noise ratio over a considerable dynamical range of the current measuring device.

d) Interpretation of the observed probe characteristics.

To minimize perturbation effects on the plasma, thin cylindrical probes appear to be the best choice (see Laframboise and Parker, 1973). If the probe diameter of cylindrical probes is smaller than the Debye length the very elegant and simple "orbital limited" probe theory applies (Eq. 5 of Laframboise and Parker, 1973). Since the orbital limited theory applies as well to ion collection as to electron collection it is desirable to measure ion and electron currents with about the same relative accuracy.

The most adverse surface effects can be overcome by sweeping the whole probe curve in a short period of time (see Geissler et al., 1972; Hoegy, 1971). The use of an On-Line computer allows one to meet requirements (b) and (c) in a very complete manner because one can sweep the probe-characteristic quickly and simultaneously record the data with a high relative accuracy over the whole range.

Figure 1 gives the analog circuitry we used to bring probe current and voltage into a form suitable for digital analysis. The probe voltage is supplied by a conventional sweep generator. The low temperature plasma which we studied required a sweep of only ± 1V. In order to make full use of the ± 5V range of the analog to digital converters (ADC) we used the voltage division scheme indicated in Fig. 1. (For cases where the probe requires a sweep over voltages exceeding 5V, one would use a different scheme.)

For the current measurement we use the fact that the difference between the output voltage of the operational amplifier (1) and its non-inverting input (= bias voltage) is R x I_{probe}. For larger probe currents and slow sweeps it is sufficient to take this difference directly from amplifier (1) with the help of amplifier (3). Amplifier (2) in Fig. 1 is essential for fast sweeps. It is used to compensate the capacity of the probe feed-through and by using the same type ampflifier it also helps to compensate (thermal) dc-drifts. The latter is essential for increased accuracy in the measurement of small (ion) currents.



Figure 1. Probe current and voltage measuring circuit.

3. EVALUATING LANGMUIR PROBE CHARACTERISTICS

Some of the problems one encounters when evaluating observed Langmuir characteristics are apparent from Fig. 2, where the solid line gives an idealized Langmuir probe characteristic which one would observe in a plasma containing ions with a mass equal to 100 electron masses. (If Fig. 2 were to be drawn for a more realistic ion mass, then the ion current would be too small to be visible.) The portion of the current indicated in Fig. 3 by a broken line is the electron current. One will obtain information on the electron energy distribution only to the degree one is able to isolate this electron current from the total observed current. The practical problems involved in deducing the electron current from the measured total current are not insignificant. The standard technique (compare Chen 1965; Brace et al., 1971) appears to be to approximate the probe characteristics below space potential (for negative χ in Fig. 2) by $I = A + VB + C.e^{-DV}$ and to pick the coefficients A through D in such a way that the mean square deviation from the experimental data is minimized. The problem with this approach is that, in general for stationary probes, the ion current is not a constant or a linear function of voltage. To illustrate this we have plotted in Fig. 4 the square of a measured probe current as a function of voltage. These data were obtained in an extended negative glow discharge (Geissler, et al., 1972). For sufficiently negative probe potential the square of the ion current is well represented by the straight line as is the electron current for high enough potentials. It can be deduced from advanced probe theory that linear extrapolation of the square of the ion current gives surprisingly accurate results up to space potential. At the point where ion and electron currents are equal (at the voltage where the observed current crosses zero) the difference between the true ion current and the value obtained from the linear representation of its square is less than .5%. At about 6kT below space potential this deviation is only about .2%. From the discussion following Eq. (6) of Laframboise and Parker (1973) one can derive the following equation for the current



Figure 2. Theoretical probe characteristics.



Figure 3. The square of an observed probe characteristic and the computed last square fit.

collected for either charge species.

$$I^2 = \left(\frac{n}{\pi}\right)^2 \left(\frac{2kT}{m} - \frac{2e(V - V_{space})}{m}\right) . A_p^2 \quad . \qquad (1)$$

A_p is the effective probe area. We write this in the form

$$I^2 = \alpha + \beta V \qquad (2)$$

with

$$\beta = (A_p)^2 \cdot \left(\frac{n}{\pi}\right)^2 \cdot \frac{2e}{m} \quad . \qquad (3)$$

Here n is the plasma density far away from the probe, e is the charge and m the mass of the attracted species. Fitting Eq. (2) to the observed electron current (see Fig. 4) determines β, which gives the density as

$$n = \pi \left(\beta \frac{m}{2e}\right)^{\frac{1}{2}} / A_p \quad . \qquad (4)$$

For electrons $e/m = 1.76 . 10^{11}$ A sec/Kg. For ions e/m will have some effective value, depending on mixing and degree of ionization. (See Appendix I.) An essential part of our work is that we treat the ion current according to Eq. (2) and not just as a linear correction.

4. THE PROCEDURE FOR ENTERING A LANGMUIR PROBE CHARACTERISTIC INTO THE COMPUTER MEMORY

As indicated in Fig. 1, the probe bias is swept linearly from negative V_{min} to positive V_{max}. The sweep is initiated by starting the sampling program. The program is started from the keyboard controlling the computer. Starting the program causes a digital to analog convertor (DAC) to issue a trigger pulse to a linear sweep generator. Optimally the DAC would provide the sweep under program control if the DAC output steps were uniform and switched without oscillating. Using a DAC for the sweep would require filtering to remove the steps in the sweep waveform. If high voltages are required for a sweep voltage, say on the order of 100V or higher, a sweep generator would still be required as DAC's are not readily available for high voltage applications. The probe voltage is held at V_{min} while in the standby mode since the resulting ion bombardment tends to keep the probe clean.

When the sweep is started, voltage and current values are sampled alternately with the same analog to digital converter by use of a multiplexer (see Fig. 2) and the resulting numbers are stored in two files in the computer. The sampling is controlled by an external clock so that there is a fixed time interval between samples. The voltage values are then interpolated so that the final voltage stored is the voltage at which the current was read. This procedure is quite accurate because the voltage sweep is linear. A four-gain programmable amplifier is used to improve the resolution of the current values as the current in the ion region is typically 10^{-2} of the current in the electron region. The multiplexer channel selection and gain selection are made simultaneously by executing a single computer instruction. A selection array is stored in the computer so that each sampled data point is sampled at a predetermined gain.



Figure 4. Communication between On-Line Computer and Experiment.

We start taking data by taking one or more trial samples with all current values sampled at the lowest gain. Typically 48 voltage-current pairs will be input (because that many points fill our display screen), The sample is then examined and each current selection word in the selection array is adjusted so that the current data will be taken at the highest gain possible without overdriving the ADC network. The gains in this system were set very close to 1, 10, and 100. Extreme care must be taken to align the DC offset controls so that the desired resolution is maintained. The actual gains of the programmable amplifier were measured and the measured values are the ones used in converting the sampled data into the final voltage and current values.

Any residual nonlinearity or offset properties which can be observed can be corrected by inputing a known function into the sampling system and digitally adjusting each data point to its actual value.

After the trial runs and (digital) adjustments were completed the actual probe data were entered in the previously described manner. The data which were entered in two's complement binary form were converted to floating point in such a way as to preserve as much accuracy as possible during data manipulation. The floating point values were then multiplied by the empirically determined gain factors which were determined during calibration runs described above. The true voltage and current values were displayed on the computer terminal and also punched on paper tape for later analysis by another computer.

The resolution was limited by the 12 bit resolution of the ADC for gains 1 and 10 of our programmable amplifier. For gain 100 the resolution was somewhat noise limited to the equivalent of a 10 bit ADC. Small changes like placing the addressable amplifier close to the experiment and using amplifiers of the low noise variety will remove this limitation.

In practice the non-ideal behavior of the analog circuitry can be largely compensated for in the following manner: after a probe curve has been entered into the computer memory one immediately enters the output of a second sweep but with the plasma turned off (or the amplifier input switched to an electrically identical mockup probe which is not in contact with a plasma). This second "characteristic" contains (1) the DC drift of the amplifier chain, (2) a small slope which is due to noncompensated input impedance in the first amplifier of the chain, and (3) the total noise of the amplifier chain. In the following we call this second characteristic the DC-offset data because DC offset is its most important contribution.

Taking DC-offset data before and after each probe characteristic gives information on how the sampling system drifts with respect to DC-offset. Generally the drift was not significant.

5. THE DATA ANALYSIS PROGRAM

This program computes the electron density and temperature and effective ion mass from the sampled data by using a least square fit to the different bias regions of the Langmuir probes.

The program has a driver that allows parameter input, data input, DC-offset calculations, data listing and data analysis by typing a one character code. Each portion of the program returns to the driver upon termination. The data to be processed must be stored on a computer file prior to running the program. A sequence of operations is necessary to analyze a data set. The sampling parameters must be entered to match the size and gain factors at which the data array was sampled. After this is done the DC-offset data are processed. A linear least square fit is calculated. Only samples taken at gain 100 are used in this procedure because for lower gains the DC-offset data consist only of zeros. The resulting linear expression is used to correct the actual probe data.

Next the probe data are processed and the data analysis program is run. The first step in analyzing the data is to remove the DC offset by subtracting the offset as described above. The probe characteristic data are now classified into three regions, the ion region, the exponential region, and the electron region. These regions are clearly defined in the process of evaluating the data.

The ion region, the most negative part of the voltage-current data, is determined by computing a least square fit to the square of the current, according to the orbital theory (Laframboise and Parker, 1973). Starting with the most negative part of the data, a fit is computed to two points, then three, four and so on until the last point deviates from the fitted curve by more than three standard deviations. This point clearly defines the limit of applicability of the orbital theory to the ion region of the data.

Now the same process is applied to the electron region, the most positive part of the data. Starting with a two point fit and the most positive voltage values, a successively larger number of points are fitted until the most negative of these points is more than three standard deviations from the fitted curve. This point defines the limit of the electron orbital region. The slope of the fit to the electron region gives the electron density n_e (see Eq. (3)) independent of electron temperature. In order to obtain the electron temperature one has to analyze the exponential region which is between the ion and electron region. In our program we use the following criteria to find the space potential which is the upper limit of this region. We guess an approximate electron temperature and calculate with the previously determined electron density the thermal electron flux ($1/4\ n\bar{v}$) which gives the probe current at space potential. That current is in general not a value sampled and the corresponding voltage is found by interpolation. If the temperature guess was not too high the next voltage point below the interpolated value is in the exponential region. The next requirement is to eliminate the ion current from the exponential region. In this region the ion current is still orbit limited and can be determined by extrapolating the curve fitted to the ion current measured at very negative voltages where no electron current is present. The difference between the sampled data and the extrapolated ion region is the true electron current in the exponential region. For sufficiently negative voltages this procedure would yield negative values. Starting from or below space potential we subtract the extrapolated ion current from the observed current until the difference is equal or smaller than three standard deviations of the systems noise. The latter is determined at the beginning of the program from the DC-offset data.

The least squares fitting to the log values of the current in the exponential region is started at floating potential, a point which is clearly in the exponential region. We include successively increasing currents into the fitting procedure up to the sample point which deviates more than three standard deviations from the fit. This determines the upper limit of the exponential fit region. The lower limit is determined in a like manner by working down from floating potential. The complete fit defines the electron temperature which could be used to iteratively update the space potential determination.

As a final step the effective ion mass is calculated from ion and electron orbital regions. (See Appendix I.)

In each region of the probe characteristics, the least fit parameters are printed under the region heading. Alpha is the vertical axis intercept, beta is the slope of the linear fit, sigma is the standard deviation, and max. deviation is the value of the point that had the greatest deviation from the fitted curve. The numbers of the data points included in the fit are also given. The DC-offset region fit produces an equation to represent the DC-offset function. That function is

$$vo_k = \alpha + \beta V_k \quad .$$

The ion and electron region data are evaluated as a linear fit to the square of the current. The exponential region fit is a linear fit to the natural log of the current. The fitted curves to the ion, exponential, and electron regions are shown in Figs. 4 and 5.



Figure 5. The electron current in the exponential region. Only the points along the solid line were used for the determination of T_e.

After the space potential, floating potential, electron density, and difference between space potential and floating potential are printed, the true electron current in the exponential region is printed point by point. It is to these current values that the exponential fit is made. Then after the exponential region is evaluated, the electron temperature and effective ion mass are printed. Finally, the percent deviation of computed electron current in the exponential region and the curve fitted to that region is printed for each point in the fitted region.

Below we give an example of the actual printout underlying Figure 5. The raw data are given at the end of Appendix II.

Langmuir Probe Analysis Program Results

```
DC OFFSET

POINTS  1 TO 25 ALPHA = -.1279E- 1 BETA= .7078E- 2
 SIGMA= .4094E- 2 MAX. DEV.= .1119E- 1

ION REGION

POINTS  1 TO 22 ALPHA= .1423E  0 BETA=-.10424E  1
 SIGMA= .6251E- 2MAX. DEV.= .1166E- 1

ELECTRON REGION

POINTS 34 TO 47 ALPHA=-.9660E   3 BETA= .1526E  5
 SIGMA= .8537E  1 MAX. DEV.= .1458E  2
SPACE POT.= .9523E- 1ELEC. DENS.= .3422E 11
VFLOAT=-.4378E- 1 POTDIF= .1390E  0

EXPONENTIAL REGION

POINT 30 ELEC. CUR.= .1508E  2
POINT 29 ELEC. CUR.= .7389E  1
POINT 23 ELEC. CUR.= .3223E  1
POINT 27 ELEC. CUR.= .1307E  1
POINT 26 ELEC. CUR.= .5167E  0
POINT 25 ELEC. CUR.= .2122E  0
POINT 24 ELEC. CUR.= .7525E- 1
POINT 23 ELEC. CUR.= .4194E- 1
POINT 22 ELEC. CUR.= .1648E- 1
POINTS 25 TO 29 ALPHA= .4243E  0 BETA= .3017E  2
 SIGMA= .2410E- 1 MAX. DEV.= .3311E- 1
EL. TEMP=     .0331 EFF. MASS=   8.1162
POINT 25 % DEV.=      .7995
POINT 26 %DEV.=     1.4728
POINT 27 %DEV.=    -8.7635
POINT 23 %DEV.=    -2.7168
POINT 29 %DEV.=     1.6553
```

6. REFERENCES

Bernstein, I. B. and I. N. Rabinowitz (1959), Theory of electrostatic probes in low-density plasma, Phys. Fluids 2, 112.

Brace, L. H., G. R. Carignan, and J. A. Findlay (1971), Evaluation of ionospheric electron temperatures by cylindrical electrostatic probes, Space Research XI, Akademie-Verlad Berlin.

Brace, L. H. and J. P. McClure (1970), Evidence for an altitude dependence in the discrepancy between rocket probe and radar backscatter observations near the magnetic equator. (Unpublished article.)

Carlson, H. C. and J. Sayers (1970), Discrepancy in electron temperatures deduced from Langmuir probes and from incoherent scatter radars, J. Geophys. Res. 75, 4883.

Chenn, F. F. (1965), Electric probes, in Plasma Diagnostic Techniques, edited by R. H. Huddlestone and S. L. Leonard, Academie, New York.

Geissler, K. H., R. A. Greenwald, and W. Calvert (1972), Measurement of plasma properties from the cross spectrum of thermal density fluctuations, Phys. Fluids 15, 96.

Hoegy, W. R. (1971), Probe and radar temperatures in an isotropic nonequilibrium plasma, J. Geophys. Res. 76, 8333.

Laframboise, J. G. (1966), Theory of spherical and cylindrical Langmuir probes in a collisionless, Maxwellian plasma at rest, University of Toronoty, UTIAS Report No. 100.

Laframboise, J. G. and L. W. Parker (1973), Probe design for orbit-limited current collection, Phys. Fluids 16, 629.

Parker, L. W. and E. C. Whipple, Jr. (1967), Theory of a satellite electrostatic probe, Annals of Physics 44, 126.

APPENDIX I: DETERMINATION OF THE AVERAGE ION MASS

To simplify the argument we assume that we have only two singly ionized ion species with masses m_1 and m_2 with densitites n_1 and n_2, respectively. This is the case for instance, in a Helium disc rge where He^+ and He_2^+ ions are present. According to Eq. (6) of Laframboise and Parker (1973), the ion saturation current collected per unit area by a thin Langmuir probe is

$$I_i = \sum_{\nu=1,2} n_\nu \cdot \left[\frac{kT}{2\pi\, m_\nu}\left(1 + \frac{e\varphi}{kT}\right)\right]^{\frac{1}{2}}. \tag{A1}$$

The extension to the general case is obvious. Quasi neutrality requires that $n_1 + n_2 = n_e$ where $n_e = n$ is the electron density. We now introduce the fractional concentration, q, of the first ion species, such that $n_1 = qn$, which implies $n_2 = (1 - q)n$. From Eq. (A1) we obtain

$$I_i = n\left[\frac{kT}{2\pi}\left(1 + \frac{e\varphi}{kT}\right)\right]^{\frac{1}{2}} \cdot \left(\frac{q}{\sqrt{m_1}} + \frac{(1-q)}{\sqrt{m_2}}\right). \tag{A2}$$

Finally we specialize to $m_1 = m$ and $m_2 = 2m$, as is suitable for a He discharge. In this case Eq. (A2) leads to

$$\left(\frac{\partial I_i^2}{\partial V}\Big/\frac{\partial I_e^2}{\partial V}\right)^{\frac{1}{2}} = \sqrt{\frac{m_e}{m}} \cdot \left(\frac{1}{\sqrt{2}} + \left(1 - \frac{1}{\sqrt{2}}\right) q\right). \tag{A3}$$

since $\partial I_i^2/\partial V = \text{BION}$ and $\partial I_e^2/\partial V = \text{BORB}$ have already been determined within our program we obtain the fraction, q, of Atomic He^+ ions from Eq. (A3) as

$$q = \left(\frac{\text{BION}}{\text{BORB}} \cdot \sqrt{4 \times 1832.3} - \frac{1}{\sqrt{2}}\right) \cdot \frac{\sqrt{2}}{\sqrt{2} - 1} \tag{A4}$$

or if we are just interested in the average ion mass in units of proton masses we obtain this from $m_{eff} = 8 - 4q$.

APPENDIX II: (a) Listing of the computer program written in Fortran II which will be accepted by relative small (old) computers.

```
C PROGRAM TO PROCESS INPUT COMMANDS
      SUBROUTINE COM
      TYPE 14
14    FORMAT(*COMMAND*,/)
12    CALL CHARIO(0,ICHAR)
      IF(ICHAR-51B)2,1,2
1     CALL INPUT
2     IF(ICHAR-57B)4,3,4
3     CALL OFFSET
4     IF(ICHAR-44B)6,5,6
5     CALL DATA
6     IF(ICHAR-54B)8,7,8
7     CALL LIST
8     IF(ICHAR-63B)10,9,10
9     CALL SAMDAT
10    TYPE 11
11    FORMAT(5HERROR,/)
      GO TO 12
      END
      SUBROUTINE INPUT
      DIMENSION MULT(50),V(50),C(50),CA(50),GAN(5)
      COMMON MULT,V,C,CA,AOFF,BOFF,SIGOFF,INDX,GAN,IOFF,DEVOFF
22    TYPE 20
20    FORMAT(*OPEN INPUT FILE*,/)
      CALL OPENIO(1,2,0,23,ISKP)
      IF(ISKP)22,21,21
21    DO 11 K = 1,INDX-1
      READ 2, 10, V(K), C(K),Y
10    FORMAT (6X,F20.12,7X,F20.12,F2.1)
11    CONTINUE
      CALL COM
      END
      SUBROUTINE SAMDAT
      DIMENSION MULT(50),V(50),C(50),CA(50),GAN(5)
      COMMON MULT,V,C,CA,AOFF,BOFF,SIGOFF,INDX,GAN,IOFF,DEVOFF
      INDX=48
      DO 10 K=1,24
10    MULT(K)=241B
      DO 11 K=25,28
11    MULT(K)=141B
      DO 12 K=29,INDX
12    MULT(K)=41B
      CALL COM
      END
C SUBROUTINE TO COMPUTE DC OFFSET FUNCTION
C TAKE DATA ON OPEN CURRENT INPUT BEFORE RUNNING THIS ROUTINE
      SUBROUTINE OFFSET
      DIMENSION MULT(50),V(50),C(50),CA(50),GAN(5)
      COMMON MULT,V,C,CA,AOFF,BOFF,SIGOFF,INDX,GAN,IOFF,DEVOFF
C COMPUTE LINEAR LEAST SQUARE FIT TO OFFSET DATA IN 100 GAIN
C REGION ONLY.
```

```
C FIND RANGE OF 100 GAIN REGION
      DO 5 I=1,INDX
5     CA(I)=C(I)
      DO 10 I=1,INDX
      IF(MULT(I)-241B)11,10,11
10    CONTINUE
11    CALL LSQR(1,I,AOFF,BOFF,SIGOFF,DEVOFF)
      IOFF=I
C CALL COMMAND PROCESSOR
      CALL COM
      END
C SUBROUTINE TO LIST INPUT DATA ARRAY
      SUBROUTINE LIST
      DIMENSION MULT(50),V(50),C(50),CA(50),GAN(5)
      COMMON MULT,V,C,CA,AOFF,BOFF,SIGOFF,INDX,GAN,IOFF,DEVOFF
C CLEAR COMPUTEK SCREEN
      TYPE 5
5     FORMAT(*&L                                            *,/)
C LIST VOLTAGE,CURRENT,CURRENT SQUARED
      TYPE 6
6     FORMAT(64HNO.        VOLTAGE                    CURRENT          C
     1URRENT SQUARED    ,/)
      DO 20 I=1,INDX-1
      CSQ=C(I)*C(I)
20    TYPE 10,I,V(I),C(I),CSQ
10    FORMAT(I2,2H  ,3F20.6,/)
      CALL COM
      END
C ROUTINE TO ANALYZE DATA
C
C THIS PROGRAM EVALUATES THE PROBE CHARACTERISTICS WHICH ARE
C MEASURED ONLINE WITH TH E EAI 640 COMPUTER
C THE FOLLOWING OPERATIONS ARE PERFORMED
C 1)INPUT AND EVALUATE THE DC OFFSET BEFORE RUNNING THIS PART
C 2)INPUT DATA NEXT
C 3)CALL THIS ROUTINE TO ANALYZE DATA
C
C THE EVALUATION IS DONE IN THE FOLLOWING STEPS
C 1)THE ION CURRENT IS FITTED TO THE ORBITAL THEORY FOR HIGH
C NEGATIVE VALUE VOLTAGES
C 2) THE ELECTRON DENSITY IS CALCULATED FROM THE HIGH
C POSITIVE VALUE VOLTAGES
C 3)THE ELECTRON DENSITY IS USED TO CALCULATE THE RANDOM
C CURRENT FROM WHICH THE SPACE POTENTIAL IS DETERMINED
C 4)THE ION CURRENT IS SUBTRACTED FROM THE MEASURED CURRENT
C TO OBTAIN THE ELECTRON CURRENT
C 5) AN EXPONENTIAL IS FITTED TO THE ELECTRON CURRENT UP TO
C SPACE POTENTIAL FROM WHICH ELECTRON TEMP. IS OBTAINED
C
C

      SUBROUTINE DATA
      DIMENSION MULT(50),V(50),C(50),CA(50),GAN(5)
      COMMON MULT,V,C,CA,AOFF,BOFF,SIGOFF,INDX,GAN,IOFF,DEVOFF
      TYPE 1
```

```
1       FORMAT(*&L
C PRINT RESULTS
        TYPE 12
12      FORMAT(9HDC OFFSET,/)
        CALL PT1(1,IOFF,AOFF,BOFF,SIGOFF,DEVOFF)
        INDXM1=INDX-1
C CORRECT MEASURED CURRENT BY SUBTRACTING THE DC OFFSET
        DO 10 K=1,INDXM1
10      C(K)=C(K)-AOFF-BOFF*V(K)
C DO ITERATIVE LEAST SQUARE FIT TO C(K)**2=A+B*V(K)
C START WITH 2 POINT FIT AND INCREASE NUMBER OF POINTS
C UNTIL LAST POINT IS MORE THAN 3*SIGOFF FROM FITTED CURVE
C THIS DEFINES THE ION REGION OF THE CURVE
        DO 20 K=1,INDXM1
20      CA(K)=C(K)*C(K)
        K=2
30      CALL LSQR(1,K,A,B,SIG,DEV)
        DIFF=A+B*V(K)-CA(K)
        IF(ABS(DIFF)-3.*SIGOFF)31,31,32
31      K=K+1
        AION=A
        BION=B
        SIGION=SIG
        DEVION=DEV
        GO TO 30
32      KION=K-1
        TYPE 40
40      FORMAT(10HION REGION,/)
C PRINT LEAST SQUARE RESULTS
        CALL PT1(1,K,AION,BION,SIGION,DEVION)
C
C DO ITERATIVE LEAST SQUARE FIT IN UPPER PART OF CURVE
C START WITH INDX-1 AND WORK DOWN TO 3*SIGOFF DEVIATION
        K=INDX-3
50      CALL LSQR(K,INDXM1,A,B,SIG,DEV)
        DIFF=A+B*V(K)-CA(K)
        OFF=15.
        IF(ABS(DIFF)-OFF)51,51,52
51      K=K-1
        AORB=A
        BORB=B
        SIGORB=SIG
        DEVORB=DEV
        GO TO 50
52      KORB=K+1
        TYPE 60
60      FORMAT(15HELECTRON REGION,/)
C PRINT RESULTS
        CALL PT1(K,INDXM1,AORB,BORB,SIGORB,DEVORB)
C CALCULATE SPACE POTENTIAL
C
C FIND CURRENT CORRESPONDING TO SPACE POT.
        SPACEI=.5*SQRT(.09424778*BORB)
C FIND POINT AT OR BELOW VALUE OF SPACEI
```

```
      DO 70 N=1,INDXM1
      K=INDXM1-N+1
      IF(SPACEI-C(K))70,71,72
70    CONTINUE
71    VSPACE=V(K)
      GO TO 73
C INTERPOLATE BETWEEN POINTS K AND K+1 TO GET SPACE POT.
72    VSPACE=V(K)+(V(K+1)-V(K))*(SPACEI-C(K))/(C(K+1)-C(K))
73    KSPACE=K
C CALCULATE ELECTRON DENSITY
      ELN=0.277E+9*SQRT(BORB)
      PRINT 80,VSPACE,ELN
80    FORMAT(11HSPACE POT.=,E10.4,12HELEC. DENS.=,E10.4,/)
C FIND FLOATING POTENTIAL
      DO 90 N=1,INDXM1
      K=INDXM1-N+1
      IF(C(K))91,92,90
90    CONTINUE
92    VFLOAT=V(K)
      GO TO 93
C INTERPOLATE TO GET FLOATING POT.
91    VFLOAT=V(K)+(V(K+1)-V(K))*(-C(K))/(C(K+1)-C(K))
93    POTDIF=VSPACE-VFLOAT
      PRINT 100,VFLOAT,POTDIF
100   FORMAT(7HVFLOAT=,E10.4,8H POTDIF=,E10.4,/)
C K=VALUE OF K CLOSE TO FLOATING POTENTIAL
C START AT THIS POINT TO FIT EXPONENTIAL CURVE.
      KFLOAT=K
C COMPUTE ELECTRON CURRENT BY EXTRAPOLATING ION REGION
C AND CORRECTING MEASURED CURRENT
      DO 104 I=1,KSPACE
      K=KSPACE-I+1
      SQION=AION+BION*V(K)
      IF(SQION)200,201,201
200   TYPE 202,K
202   FORMAT(5HPOINT,I3,10H NEG. ROOT,/)
      GO TO 104
201   CA(K)=C(K)+SQRT(AION+BION*V(K))
C IF ELECTRON CURRENT IS LESS THAN 3*SIGOFF DO NOT TAKE LOG
      IF(CA(K)-3.*SIGOFF)101,101,102
C PRINT ELECTRON CURRENT OF EACH POINT AS OBTAINED
102   PRINT 110,K,CA(K)
110   FORMAT(6HPOINT ,I2,12H ELEC. CUR.=,E10.4,/)
      CA(K)=ALOG(CA(K))
104   CONTINUE
C USE LEAST SQUARE FIT TO THE LOG OF THE CURRENT IN THE
C EXPONENTIAL REGION
C START AT KFLOAT AND WORK UP TO KSPACE UNTIL A 3*SIGOFF
C DEVIATION IS OBTAINED
101   KUPPER=KFLOAT+1
103   CALL LSQR(KFLOAT,KUPPER,A,B,SIG,DEV)
      OFF1=.05
      IF(ABS((A+B*V(KUPPER)-CA(KUPPER)))-OFF1)120,120,121
120   KUPPER=KUPPER+1
```

```
      AEXP=A
      BEXP=B
      SIGEXP=SIG
      DEVEXP=DEV
      GO TO 103
C WHEN 3 SIGOF POINT IS REACH SAVE THAT POINT AND WORK
C BACK DOWN TO A 3 SIGOFF DEVIATION AT THE OTHER END OF THE
C CURVE
121   KUPPER=KUPPER-1
122   KFLOAT=KFLOAT-1
      CALL LSQR(KFLOAT,KUPPER,A,B,SIG,DEV)
      IF(ABS(A+B*V(KFLOAT)-CA(KFLOAT))-OFF1)130,130,131
130   AEXP=A
      BEXP=B
      SIGEXP=SIG
      DEVEXP=DEV
      GO TO 122
131   KFLOAT=KFLOAT+1
C PRINT RESULTS
      TYPE 140
140   FORMAT(11HEXP. REGION,/)
      CALL PT1(KFLOAT,KUPPER,AEXP,BEXP,SIGEXP,DEVEXP)
C CALCULATE ELECTRON TEMPERATURE
      TEMP=1./BEXP
C CALCULATE EFFECTIVE MASS
      EFFM=-BORB/(EION*1836.3)
      TYPE 150,TEMP,EFFM
150   FORMAT(9HEL. TEMP=,F10.4,11H EFF. MASS=,F10.4,/)
C CALCULATE % DEVIATION FROM FITTED CURVE
      DO 160 K=KFLOAT,KUPPER
      DIFF=100.*(AEXP+BEXP*V(K)-CA(K))/ABS(CA(K))
160   PRINT 170,K,DIFF
170   FORMAT(5HPOINT,I3,7H %DEV.=,F10.4,/)
C CALL COMMAND PROCESSOR
      CALL COM
      END
C SUBROUTINE TO COMPUTE LINEAR LEAST SQUARE FIT THRU I-M POINTS
      SUBROUTINE LSQR(M,I,ALPHA,BETA,SIGMA,DELMAX)
      DIMENSION MULT(50),V(50),C(50),CA(50),GAN(5)
      COMMON MULT,V,C,CA,AOFF,BOFF,SIGOFF,INDX,GAN,IOFF,DEVOFF
C CLEAR SUMS
      SUMVOL=0.
      SUMCUR=0.
      SUMVSQ=0.
      SUMVCR=0.
      DO 10 K=M,I
C SUM CURRENT
      SUMCUR=SUMCUR+CA(K)
C SUM VOLTAGE
      SUMVOL=SUMVOL+V(K)
C SUM VOLTAGE SQUARED
      SUMVSQ=SUMVSQ+V(K)*V(K)
C SUM VOLTAGE TIMES CURRENT
10    SUMVCR=SUMVCR+V(K)*CA(K)
```

```
C COMPUTE LINEAR EQUATION COEFFICIENTS ALPHA,BETA IN
C CA(K)=ALPHA+BETA*V(K)
      ALPHA=(SUMCUR-SUMVCR*SUMVOL/SUMVSQ)/((I-M+1)-SUMVOL*
     1SUMVOL/SUMVSQ)
      BETA=(SUMVCR-ALPHA*SUMVOL)/SUMVSQ
C COMPUTE RMS CURRENT(SIGMA) AND MAX DEVIATION
      DELMAX=0.
      SUMDEL=0.
      DO 20 K=M,I
      DEL=(CA(K)-ALPHA-BETA*V(K))*(CA(K)-ALPHA-BETA*V(K))
      IF(DEL-DELMAX)20,20,19
19    DELMAX=DEL
20    SUMDEL=SUMDEL+DEL
      SIGMA=SQRT(SUMDEL/(I-M+1))
      DELMAX=SQRT(DELMAX)
      RETURN
      END
C ROUTINE TO PRINT LEAST SQUARE FIT RESULTS
      SUBROUTINE PT1(L,IU,A,B,SIG,DEV)
      DIMENSION MULT(50),V(50),C(50),CA(50),GAN(5)
      COMMON MULT,V,C,CA,AOFF,BOFF,SIGOFF,INDX,GAN,IOFF,DEVOFF
      TYPE 10,L,IU,A,B,SIG,DEV
10    FORMAT(6HPOINTS,I3,3H TO,I3,7H ALPHA=,E10.4,6H BETA=,
     1E10.4,/,7H SIGMA=,E10.4,11H MAX. DEV.=,E10.4,/)
      RETURN
      END
```

b) Example of Langmuir Probe data. These are the data pertaining to Figs. 4 and 5.

NO.	VOLTAGE	CURRENT	CURRENT SQUARED
1	-.774048	-.964107	.929503
2	-.744374	-.946344	.895566
3	-.715462	-.934572	.873424
4	-.685811	-.917809	.842374
5	-.656159	-.906538	.821812
6	-.626747	-.882786	.779311
7	-.597334	-.866022	.749995
8	-.567683	-.849260	.721243
9	-.538031	-.837490	.701389
10	-.508380	-.821227	.674414
11	-.478728	-.797476	.635969
12	-.449316	-.779714	.607954
13	-.419904	-.754464	.569217
14	-.390252	-.742195	.550853
15	-.360601	-.719942	.518316
16	-.330949	-.691200	.477757
17	-.301537	-.675434	.456211
18	-.272124	-.643196	.413701
19	-.242473	-.622940	.388054
20	-.213060	-.597690	.357233
21	-.183409	-.565453	.319737
22	-.153518	-.530723	.281666
23	-.124106	-.477020	.227548
24	-.094694	-.413833	.171253
25	-.065042	-.244822	.059938
26	-.035630	.093910	.008819
27	-.005978	.921333	.848854
28	.023673	2.879378	8.290816
29	.053325	7.092507	50.303658
30	.082976	14.839770	220.218775
31	.112150	24.655720	607.904526
32	.141562	32.870088	1080.442717
33	.171213	39.689527	1575.258582
34	.200865	45.072354	2031.517121
35	.230277	50.331886	2533.298728
36	.259929	54.723117	2994.619564
37	.289580	58.752700	3451.379779
38	.319232	62.523959	3909.245503
39	.348883	66.036903	4360.872511
40	.378296	69.394849	4815.645035
41	.407947	72.546143	5262.942923
42	.437360	75.645773	5722.283047
43	.466772	78.538754	6168.335846
44	.496423	81.276734	6605.907427
45	.526075	84.014713	7058.472075
46	.555726	86.649376	7508.114363
47	.585378	89.180691	7953.195611